```
(*
                            CS 51 Lab 19
                Conway's Game of Life Cellular Automaton
 *)
(*
                               SOLUTION
 *)
module G = Graphics ;;

(* Automaton parameters *)
let cGRID_SIZE = 100 ;;      (* width and height of grid in cells *)
let cSPARSITY = 5 ;;         (* inverse of proportion of cells initially live *)
let cRANDOMNESS = 0.00001 ;; (* probability of randomly modifying a cell *)

(* Rendering parameters *)
let cCOLOR_LIVE =   G.rgb  93  46  70 ;;  (* color to depict live cells *)
let cCOLOR_DEAD =   G.rgb 242 227 211 ;;  (* background color *)
let cCOLOR_LEGEND = G.rgb 173 106 108 ;;  (* color for textual legend *)
let cSIDE = 8 ;;             (* width and height of cells (in pixels) *)
let cRENDER_FREQUENCY = 1    (* how frequently grid is rendered (in ticks) *) ;;

(* Font specification for rendering the legend. OCaml font handling is
   platform dependent, so we leave this as `None`, but we provide the
   functionality for use in the solution set. Feel free to play with
   this or leave it as is. *)
let cFONT = Some "-adobe-times-bold-r-normal--34-240-100-100-p-177-iso8859-9"


(*----------------------------------------------------------------------
  The Game of Life
 *)

(*......................................................................
  Game of Life states and updates
 *)

(* We take the opportunity to abstract the cell states as an
   enumerated type *)
type life_state =
  | Live
  | Dead ;;

(* flipped state -- Returns the opposite state to `state`. *)
let flipped (state : life_state) : life_state =
```

```
45    match state with
46    | Live -> Dead
47    | Dead -> Live ;;
48
49  (* offset index off -- Returns the `index` offset by `off` allowing
50     for wraparound. *)
51  let rec offset (index : int) (off : int) : int =
52    if index + off < 0 then
53      cGRID_SIZE - (offset ~-index ~-off)
54    else
55      (index + off) mod cGRID_SIZE ;;
56
57  (* life_update grid i j -- Returns the updated value for cell at `i,
58     j` in the grid based on rules of Conway's Game of Life. In this
59     implementation, after updating as per the standard GoL update rule,
60     a cell's state is then flipped with probability given by
61     cRANDOMNESS. *)
62  let life_update (grid : life_state array array) (i : int) (j : int)
63                  : life_state =
64
65    (* We give a few variant update rules for the Game of Life. These
66       lists are indexed by adjacency count to give the generated cell
67       status.
68                             0     1     2     3     4     5     6     7     8
69                             |     |     |     |     |     |     |     |     |
70                             v     v     v     v     v     v     v     v     v   *)
71    (* B3/S23: Conway's original game of life *)
72       let dead_update = [Dead; Dead; Dead; Live; Dead; Dead; Dead; Dead; Dead] in
73       let live_update = [Dead; Dead; Live; Live; Dead; Dead; Dead; Dead; Dead] in
74
75    (* B3/S12345: https://conwaylife.com/wiki/OCA:Maze
76       let dead_update = [Dead; Dead; Dead; Live; Dead; Dead; Dead; Dead; Dead] in
77       let live_update = [Dead; Live; Live; Live; Live; Live; Dead; Dead; Dead] in
78     *)
79    (* B36/S125: https://conwaylife.com/wiki/OCA:2×2
80       let dead_update = [Dead; Dead; Dead; Live; Dead; Dead; Live; Dead; Dead] in
81       let live_update = [Dead; Live; Live; Dead; Dead; Live; Dead; Dead; Dead] in
82     *)
83    (* B3678/34678: https://conwaylife.com/wiki/OCA:Day_%26_Night
84       let dead_update = [Dead; Dead; Dead; Live; Dead; Dead; Live; Live; Live] in
85       let live_update = [Dead; Dead; Dead; Live; Live; Dead; Live; Live; Live] in
86     *)
87
88    let neighbors = ref 0 in
89    for i' = ~-1 to ~+1 do
90      for j' = ~-1 to ~+1 do
```

```
      let i_ind = offset i i' in
      let j_ind = offset j j' in
      neighbors := !neighbors
                   + match grid.(i_ind).(j_ind) with
                     | Live -> 1
                     | Dead -> 0
    done
  done;
  (* determine new state based on neighbor count *)
  let new_state =
    match grid.(i).(j) with
    | Live -> List.nth live_update (!neighbors - 1)
    | Dead -> List.nth dead_update !neighbors in
  (* randomly flip an occasional cell state *)
  if Random.float 1.0 <= cRANDOMNESS then
    flipped new_state
  else
    new_state ;;

(* life_random_state () -- Returns a random cell state. *)
let life_random_state () =
  let states = [Live; Dead] in
  List.nth states (Random.int (List.length states))

(*.....................................................................
  Making the Game of Life
 *)

(* GoL automaton specification *)

module LifeSpec : (Cellular.AUT_SPEC
                   with type state = life_state) =
  struct
    type state = life_state
    let initial : state = Dead
    let grid_size : int = cGRID_SIZE
    let random_state = life_random_state
    let update = life_update
    let name : string = "Conway's Life"
    let cell_color (cell : state) : G.color =
      match cell with
      | Live -> cCOLOR_LIVE
      | Dead -> cCOLOR_DEAD
    let side_size : int = cSIDE
    let legend_color : G.color = cCOLOR_LEGEND
    let font : string option = cFONT
```

```
137     let render_frequency : int = cRENDER_FREQUENCY
138   end ;;
139
140 (* GoL automaton *)
141
142 module Aut = Cellular.Automaton (LifeSpec) ;;
143
144 (*.....................................................................
145   Running the automaton and displaying the results
146  *)
147
148 (* main seed -- Runs the automaton using the provided random `seed`
149    (for replicability), displaying updates to the graphics window. *)
150 let main seed =
151
152   Random.init seed;      (* initialize randomness *)
153   Aut.graphics_init ();  (* ... and graphics window *)
154
155   let initial_grid = Aut.random_grid () in
156   Aut.run_grid initial_grid ;;
157
158 (* Run the automaton with user-supplied seed *)
159 let _ =
160   if Array.length Sys.argv <= 1 then
161     Printf.printf "Usage: %s <seed>\n  where <seed> is integer seed\n"
162                   Sys.argv.(0)
163   else
164     main (int_of_string Sys.argv.(1)) ;;
165
```